
Abstract: We first briefly describe an algebraic model of classical and modal many-valued logics due to the authors and 
introduced in previous works. A similar approach, also using Computer Algebra techniques (Gröbner bases) and oriented 
to perform effective calculus in a continuous logic: Minimal Polynomial Logic (MPL) is presented. The implementation 
has been developed in the Computer Algebra System Maple. The possibility to perform knowledge extraction and to 
check consistency in Rule Based Expert Systems (RBES) whose underlying logic is MPL, has also been explored. The ar-
ticle is illustrated with examples of very simple RBES. 

INTRODUCTION 

Motivation 

 Since the early nineties, we have been developing an 
algebraic approach to knowledge extraction and consistency 
checking in Rule Based Expert Systems (RBES) using resi-
due class rings and Gröbner bases (GB). The underlying 
logic can be either classical Boolean logic or many-valued 
modal logic. 
 The effective computations are carried out using Com-
puter Algebra Systems and the implementations are surpris-
ingly brief.  
 Different applied works have been successfully carried 
out, mainly in Medicine.  
 This paper studies the possibility to extend this treatment 
to the continuous logic that is, possibly, most similar: Mini-
mal Polynomial Logic (MPL). 

Some Introductory Notes About RBES 

 Expert Systems (ES) are a particular type of Knowledge 
Based Systems. An ES can be considered as a system of 
automated deduction from the information provided by ex-
perts. ES basically consist of three components:  
• a Knowledge Base (KB), the construction of which 

requires to choose an strategy leading to a model of 
knowledge gathering, the aim of which is to orderly 
collect and symbolically represent the available in-
formation;  

• an Inference Engine (IE) which both verifies the con-
sistency of the KB and extracts consequences auto-
matically;  

• and, possibly, an interactive Graphic Users Interface 
(GUI), for people not necessarily familiar with the 
logical and mathematical details of the system con-
struction. 

 

*Address correspondence to this author at the Algebra Department, Facultad 
de Educacion, Universidad Complutense de Madrid, c/Rector Royo Villa-
nova s/n, 28040-Madrid, Spain; E-mail: eroanes@mat.ucm.es 

The ES treated here are RBES. In this case, the strategy lead-
ing to its logic formulation consists of translating all the in-
formation formulated in natural language into a set of pro-
positional logic formulae, called “production rules”, of the 
form δ → φ, where δ is a literal1 or a conjunction of literals, 
and φ is a literal or a disjunction of literals. If the underlying 
logic is classical Boolean logic, an example of production 
rule is: 

x1 ∧ x2 ∧ ¬x3 ∧ ¬x5 → x9 ∨ x13 
that is read as: “if x1 and x2 and not x3 and not x5 hold, then x9 
or x13 holds”. 
 The set of all literals in the KB contains a distinguished 
subset called “the set of potential facts”. The “potential 
facts”, are the literals that do appear in the left side of one 
production rule (at least) and such that neither them nor their 
contraries appear in the right side of any production rule2, 
together with their contraries. For instance, the whole set of 
potential facts can be something like: 

{x1, ¬x1, x2, ¬x2, x5, ¬x5, x8, ¬x8, x16, ¬x16} 
 That a given set of facts (i.e., a subset of the set of poten-
tial facts that is stated as true) is “consistent” means that, 
from each pair formed by a potential fact and its contrary, 
like (x5,¬x5), only one literal, for instance ¬x5, is chosen. 
That it is “maximal” means that such a choosing must be 
done in all pairs. 
 Some KB also contain other items known as “integrity 
constraints” (IC). An IC is a logical formula that translates 
some experts' opinions about the convenience of stating that 
two or more potential facts, for instance x8 and x16, ought not 
to occur simultaneously. 
 A (consistent) set of potential facts, together with the set 
of production rules and IC, constitute the KB. 

                                                
1A “literal” is a propositional variable of the underlying logic or its negation. 
2In case the underlying logic is modal many-valued, modal connectives also have to be 
taken into account here. 

 Intuitively, “forward firing” is the way in which, what is 
stated as “true”, propagates through the RBES. It corre-
sponds to the logic concept of tautological consequence3. 
 The IE is a an automated tool (in our case, a program in 
the Computer Algebra System -CAS), that verifies consis-
tency (that is, checks that the system does not lead to contra-
dictions) and draws consequences from the information 
contained in the KB (that corresponds to the logical concept 
of “tautological consequence”). 
 A RBES is inconsistent when a consistent set of facts, 
such that any formulae formed with the variables contained 
in the KB is a tautological consequence of the information 
contained in the RBES, exists. In particular, contradiction 
would be a tautological consequence of such information.  
Example 1: Let us consider the trivial RBES formed by the 
three rules: 
R1: x ∧ y → v 
R2: v → z 
R2: w →¬z 
 The potential facts are {x,¬x,y,¬y,w,¬w}. If the given set 
of facts is {x,w}, only ¬z is obtained by forward firing (no 
inconsistency is found starting with this set of facts). But if 
{x,y,¬w} are given as facts (it is a consistent set of facts), 
both z and ¬z are obtained by forward firing, so the RBES is 
inconsistent. 
 In case the underlying logic is many-valued, two diffe-
rent types of consistency can be distinguished [1].  

An Algebraic Model of RBES Whose Underlying Logic is 
Classical Boolean or Modal Many-Valued 

 The next theorem relates tautological consequence in 
logic with an ideal membership problem in polynomial ring 
theory. The theorem is applied to verify consistencies in, and 
draw consequences from, the KB. 
 The logical and mathematical ideas behind the GB-based 
IE can be found in the mathematical model of logic and 
RBES in [2,3], that are based on previous works for rewrit-
ing classical Boolean [4,5] and many-valued logics [6]. 
 For instance, in classical Boolean logic, the polynomial 
translations of the logic connectives are 

x1∨x2 = x1 + x2 - x1 x2 
x1∧x2 = x1x2 

¬x1 = 1-x1 
 (these are the same polynomials George Boole used in his 
seminal works on logic).  
 Similarly, the polynomial expressions corresponding to 
the basic logical formulae in Lukasiewicz's three-valued 
logic (if 2 is assigned to “true”, 1 to “undetermined” and 0 to 
“false”, and computations are performed in Z3) are: 

¬x1 = 2x1+2 
□x1 = x1

2+2x1 

                                                
3A logical formula, α, is a “tautological consequence” of a set S of formulae, denoted 
S⊨ α, if and only if, whenever all formulae in S are true, α is also true. 

◇x1 = 2x1
2 

x1∨x2 = x1
2x2

2+x1
2x2+x1x2

2+2x1x2+x1+x2 
x1∧x2 = 2x1

2x2
2+2x1

2x2+2x1x2
2+x1x2 

(where □ means “necessarily” and ◇ means “possibly”). 

Theorem 1 

 A logical formula, α, is a tautological consequence of the 
set of formulae M={A1,A2,..., Am} in a certain p-valued modal 
logic (where p is a prime number) if and only if the polyno-
mial translation4 of ¬α in the ring Zp[x1, x2, ..., xn]/I (where5 
I=<x1

p-x1,x2
p-x2,...,xn

p-xn>), denoted Pol(¬α), belongs to the 
(polynomial) ideal <Pol(¬A1),Pol(¬A2),..., Pol(¬Am)>. 
 The effective methods for calculating normal forms (NF), 
i.e., reduction modulo ideals, in polynomial rings and char-
acterizing ideals via some very special bases (GB) are due to 
Bruno Buchberger [7]. Implementations of GB and NF are 
nowadays included in most CAS. 
 We usually denote by K the polynomial ideal generated 
by the polynomial translations of the negation of the ele-
ments of a consistent subset of the set of potential facts. 
Similarly, J is the ideal generated by the polynomial transla-
tions of the negation of the production rules and IC of the 
RBES. We shall use this notation hereinafter. 
 So we have an effective method what can decide whether 
a logical formula, α, can be obtained by forward firing from 
a consistent set of facts and the production rules and IC of 
the RBES or not: 

a logical formula, α, can be obtained by forward firing  
from a consistent set of facts and the rules and IC in the 

RBES ⇔ 
⇔ NF(Pol(¬α),K+J)=0 

(in the ring Zp[x1, x2, ..., xn]/I). 
 We can alternatively check with the CAS whether 
NF(Pol(¬α),I + K + J) is 0 or not (in the ring Zp[x1, x2, ..., 
xn]).  
 We ask whether Pol(¬α) belongs to I+K+J or not in the 
ring Zp[x1, x2, ..., xn], instead of asking whether Pol(¬α) 
belongs to K+J or not in the ring Zp[x1, x2, ..., xn]/I, because 
most CAS cannot perform calculations in residue class rings. 
 And we also have an effective method that can decide 
whether a RBES is inconsistent or not. With the notation 
above, 

a consistent set of facts leads to inconsistency ⇔ 
⇔ GB(J+K)=<1> 

(in the ring Zp[x1, x2, ..., xn]/I)6, that can be decided in the 
CAS by checking whether GB(I + K + J) is [1] or not (in the 
ring Zp[x1, x2, ..., xn]), because, if this basis is [1], from 
Theorem 1, any formula is a tautological consequence of the 
information provided by the KB (which means inconsis-
tency). 

                                                
4The polynomial translation of the logic connectives will depend on p and on the logic 
chosen. For instance, Lukasiewicz's three-valued logic and Kleene's three-valued logic 
lead to different polynomial expressions. 
5“<” and “>” are used to represent the ideal generated by the elements between those 
symbols. 
6An ideal is the whole ring if and only if its GB is [1]. 

A BRIEF OVERVIEW OF MPL 

 MPL [8,9] is a generalization of classical propositional 
logic allowing truth values in the closed interval [0,1]. The 
polynomials 

x1∨x2=1-(1-x1)(1-x2) 
x1∧x2=x1x2 

¬x1=1-x1 
fit the truth-table entries of the ordinary logic connectives, 
and are the ones used by the “lowest degree polynomial 
logic”, denoted PL0 (note that these polynomials are the 
same as those used in the previous section).  
 Now MPL is obtained from PL0 the following way 
(Definition 3 of [8]): 
 “Given a propositional formula e, its MPL version em is 
obtained from the PL0 version ep by distributing + over × 
throughout and then substituting subexpressions of the form 
xi

k (with k>1) with xi. This substitution will be sometimes be 
denoted (•)m.” 
 Truth values in MPL can be interpreted either as 
probabilities of assertions being true, as in Nilsson’s 
probabilistic logic or, also, as the degrees of truth of those 
assertions, as in fuzzy logic.  
 As in Boolean logic, e⊨e’ is trivially equivalent to e→e’. 
It can be proved (Corollary 7 of [8]) that e⊨e’ if and only if 
em≡(eme’m)m. It can also be proved that e⊨e’ if and only if 
em≤e’m for any truth values in [0,1] of the propositional vari-
ables (Theorem 11 of [8]). 
 The theoretical details about MPL can be found in [8,9] 
(we shall only deal here with an straightforward algebraic 
approach and its implementation in a CAS).  

APPROACHING MPL FROM ALGEBRA AND 
COMPUTER ALGEBRA 

 Let us observe that the substitution (•)m introduced in 
Definition 3 of [8] (quoted above) can be interpreted as the 
NF of the PL0 version of the propositional formula (a poly-
nomial of7 R[x1,...,xn]) modulo the ideal <x1

2-x1,..., xn
2-xn>.  

 Therefore, from an algebraic point of view [10], it is 
straightforward and trivial to consider the MPL version of 
the propositional formulae as polynomials lying in the resi-
due class ring  

R[x1,...,xn]/I ; I=<x1
2-x1,...,xn

2-xn> 
 Nevertheless, there is a quantitative difference w.r.t. the 
model of classical Boolean and many-valued logic that we 
had previously developed. Let us detail it.  

THE ALGEBRAIC APPROACH TO MPL 
COMPARED TO THE ALGEBRAIC APPROACH TO 
CLASSICAL BOOLEAN LOGIC 

 In the Boolean case (and also in the p-valued case, 
substituting all the 2 by p in the definitions of the ring and of 
ideal I), the residue class ring  

Z2[x1,...,xn]/I ; I=<x1
2-x1,...,xn

2-xn> 

                                                
7They could also be considered as polynomials of Z[x1,..., xn], although we shall con-
sider their real roots and shall substitute x by real values. 

is a polynomial model where ideals of the logic Boolean 
algebra do correspond one-to-one with the ideals of the po-
lynomial Boolean algebra (that are the same as the ideals of 
the polynomial Boolean ring). Therefore, it is not important 
which expression of the polynomial translation of a formula 
is chosen (i.e., which generator of the polynomial ideal8 gen-
erated by a polynomial translation of the formula is chosen).  
 Summarizing, in the Boolean case we have the beautiful 
diagram in Fig. (1). 
 
 
 
 
 
 
Fig. (1). Isomorphisms relating the propositional Boolean algebra 
and the polynomial residue class ring. 
 

Example 2 

 The polynomial translation of the formula x1∧x1 (of clas-
sical Boolean logic) in the residue class ring Z2[x1,...,xn]/I is 
x1

2=x1. And there is no inconvenience in considering, e.g., 
x1

7 as a polynomial translation of x1∧x1 in this residue class 
ring: for any truth value in {0,1}, it is the same to consider x1 
or x1

7. 
 Nevertheless, when we try to do something similar for 
MPL, R[x1,...,xn]/I ; I=<x1

2-x1,...,xn
2-xn> is not a polynomial 

model of MPL in the same sense, with a one-to-one corres-
pondence between ideals, but a residue class ring where only 
a certain polynomial translation of each logic formula (the 
canonical representative of the residue class) behaves as the 
logic formula that it translates! 

Example 3 

 Let us consider the same formula of Example 2. The po-
lynomial translation of the formula of MPL x1∧x1 in 
R[x1,...,xn]/I would be x1. Despite the fact that in that residue 
class ring, e.g. x1

7=x1
2=x1, x1 and only x1 behaves as x1∧x1 

for all values in the closed interval [0,1]. 

EFFECTIVE COMPUTATIONS IN MPL USING 
COMPUTER ALGEBRA TECHNIQUES 

 The NF of a polynomial in R[x1,...,xn] modulo the ideal 
<x1

2-x1,..., xn
2-xn>, can be effectively computed in any CAS 

including the possibility to perform GB computations, like 
Maple, Mathematica, CoCoA…  
 Let us observe that the polynomials of MPL are reduced 
modulo ideal I, and therefore an upper bound of its degree in 
each variable is 1 (and, consequently, an upper bound of its 
total degree is the number of propositional variables consid-
ered).  

MAPLE IMPLEMENTATION 

 We can easily implement this idea in a CAS, for instance, 
in Maple. 
                                                
8All ideals in these rings are principal. 

 We have to begin loading the GB package and defining 
the sequence of propositional variables, for instance x1,...,x10: 
> with(Groebner): 

> SV:=x[i] $ i=1..10: 

 Then we can define ideal I (renamed iI, as I is a reserved 
word in Maple), using an auxiliary function, denoted fu (in 
fact, a GB of the ideal, w.r.t. pure lexicographic order, is 
directly assigned to iI): 
> fu:=var->var^2-var: 

> iI:=Basis(map(fu,[SV]),plex(SV)): 

 Now we are ready to define the basic logic connectives:  
> `NEG`:=(m::algebraic)-> 1-`m`: 

> `&AND`:=(m::algebraic, 

           n::algebraic) 

 -> NormalForm(expand(`m`*`n`),  

    iI, plex(SV)): 

and the “derived” ones: 
> `&OR`:=(m::algebraic,  

          n::algebraic) 

 -> NormalForm(NEG(NEG(`m`)  

    &AND NEG(`n`)),  

    iI, plex(SV)): 

> `&IMP`:=(m::algebraic, 

           n::algebraic)  

 -> NormalForm(NEG(`m`) &OR `n`, 

    iI, plex(SV)): 

> `&XOR`:=(m::algebraic, 

           n::algebraic)  

 -> NormalForm((NEG(`m`) &AND  

     `n`) &OR (`m` &AND NEG(`n`)),  
    iI, plex(SV)): 

> `&IFF`:=(m::algebraic, 

           n::algebraic)  

 -> NormalForm((`m` &IMP `n`)  

    &AND (`n` &IMP `m`),  

    iI, plex(SV)): 

(the binary ones will be input to Maple in infix mode) and to 
define tautology and contradiction as constants: 
> t:=1: 

> c:=0: 

 Let us observe that the reductions modulo iI performed 
by command NormalForm above are computed conside-
ring the order for variables of sequence SV and the order for 
monomials “pure lexicographic”. 
 The restriction of the MPL connectives to the Boolean 
truth values {0,1} return the same truth-values as the Boolean 
connectives, as shown in Fig. (2). 

 
 
 
 
Fig. (2). MPL binary connectives restricted to {0,1} (computed with 
Maple’s implementation of MPL). 

 

SYMBOLIC EXAMPLES 

 The CAS can perform computations with non-assigned 
variables, so we can obtain and compare polynomial expres-
sions of formulae (and consequently compare the logical 
formulae themselves). 

Example 4 

 For instance, we can check that ∨ is associative (observe 
that this is not formally a proof: we only check it for proposi-
tional variables x[1], x[2] and x[3]; nevertheless it will also 
hold if x[1], x[2] and x[3] are substituted by any polynomial 
expressions, so it will be true in general)9: 
> x[1] &AND (x[2] &AND x[3]); 

 x1x2x3 

> (x[1] &AND x[2]) &AND x[3]; 

 x1x2x3 

Example 5 

 Check idempotency of ∨: 
> x[1] &OR x[1]; 

    x1 

Example 6 

 Check one of the de Morgan laws: 
> NEG(x[1] &AND x[2]); 

   1-x1x2 

> NEG(x[1]) &OR NEG(x[2]); 

  1-x1x2 

Example 7 

 Unlike what happens in many-valued modal logics, 
x1∧¬x1 is a contradiction: 
> x[1] &AND NEG(x[1]); 

 0 

 The previous examples were really simple and could be 
easily performed by hand. For complex or long examples, 
using a computer is highly preferable, and this simple Maple 
implementation (or a similar one in another CAS) could the-
refore be a very convenient tool.  
 Let us underline that the formulae that can be handled in 
this approach are completely general (for instance, implica-
tions are not restricted to Horn clauses, like in some logic 
languages and programs). 

                                                
9In Maple, statement separator “;” produces echo, meanwhile “:” produces no echo. 

NUMERICAL EXAMPLES 

 The CAS can also perform numeric computations. There-
fore we can calculate the truth value of a logical formula or 
the truth value of a propositional variable in a formula for 
the formula to have a certain truth value. 

Example 8 

 Let P be the logical formula  
x[1] ∧ x[2] → x[3]. 

 If the truth values of x[1], x[2] and x[3] are 0.3, 0.75 and 
0.8 (respectively), we can easily compute the truth value of P 
substituting all the ocurrences of these propositional vari-
ables by their numerical values (what command subs does 
in Maple): 
> P:=(x[1] &AND x[2]) &IMP x[3]; 

> subs(x[1]=0.3,x[2]=0.75,x[3]=0.8, 

       P); 

0.9550 

 Moreover, given the values of all but one propositional 
variables, we can compute the truth value of the other pro-
positional variable in order the formula to have a given truth 
value. 

Example 9 

 Let us consider the formula of Example 8. If the truth 
values of x[1] and x[3] are 0.3 and 0.8 (respectively), then 
we can compute the truth value of x[2] for the logical for-
mula P: x[1] ∧ x[2] → x[3] to have truth value 0.955 using 
command fsolve (numeric solve) : 
> fsolve(  

    subs(x[1]=0.3,x[3]=0.8,P)=0.9550, 

         x[2]); 

            0.7500000000 
so the solution is: x[2] should have truth value 0.75 (this 
procedure is the converse of Example 8). 
 Nevertheless, it is not always possible to find such a solu-
tion, as shown in Example 10. 

Example 10 

 Let us consider, again, the simple formula P of Example 
8 for the truth values of x[1] and x[3] being 0.3 and 0.8 (re-
spectively). Then, if we try to compute the truth value of 
x[2] for the logical formula P to have truth value 0.2:  
> fsolve(  

    subs(x[1]=0.3,x[3]=0.8,P)=0.2, 

         x[2]); 

            13.3333333 

is obtained, a value that is outside interval [0,1] and there-
fore not a possible truth value in MPL. 
 Anyway that a certain formula cannot reach a certain 
truth value is not a critical drawback. For instance, in Lu-
kasiewicz’s three-valued logic, no value of the propositional 
variables q and r make the logical formula ◇q ∨ ◇r undeter-
mined. 

 Finally, we can also ask the CAS to compute when a for-
mula has a truth value fulfilling an inequality. 

Example 11 

 If the truth values of x[1] and x[3] are 0.3 and 0.8 (re-
spectively), for which truth values of x[2] does formula P of 
Example 8 have a truth value strictly greater than 0.95?  
> solve(  

   subs(x[1]=0.3,x[3]=0.8,P)>0.95,  
   x[2])  

  intersect RealRange(0,1); 

     RealRange(0, 1) intersect 
     RealRange(-infinity, 
               Open(0.8333333333)) 

that is, interval [0,0.8333333333…). Let us underline that we 
have used the non-numeric solving command solve instead 
of fsolve in this case. 

A COMPARISON WITH KLEENE OR 
LUKASIEWICZ THREE-VALUED LOGIC 
CONNECTIVES 

 We have computed with Maple the truth tables of ∧ and 
∨ in MPL, restricting the input values to {0,½,1} (see Fig. 3). 
If 0, ½, 1 represented false, undertermined/undecided and 
true (respectively), the truth table obtained in Kleene or Lu-
kasiewicz three-valued logics would have been the same, 
except the central line, that would have been: ½, ½, ½, ½ 
 
 
 
 
 
 
 
 
 
 
 
 
 
Fig. (3). MPL binary connectives restricted to {0,1/2,1} (computed 
with Maple’s implementation of MPL). 

 
 The ¼ and ¾ would be the expected answers if we were 
thinking of probabilities and x[1] and x[2] were independent 
events.  

DESIGNING RBES WHICH UNDERLYING LOGIC IS 
MPL 

 Let us observe that we are going to consider the most 
general case, where: 

• any value in [0,1] can be assigned as truth value of a 
fact, rule and IC, 

• the variables that are not potential facts (i.e., that ap-
pear in consequents) can have truth values ranging in 
[0,1] too. 

Note that in our previous works based on the theory for 
many-valued modal logics in [2,3], we have always consid-
ered that the rules and IC, as well as the given facts are 
stated as “true” (never as undertermined/undecided), al-
though the forward reasoning is performed according to the 
many-valued modal logic chosen (modal connectives can 
appear within the antecedents and the consequents of the 
rules). 

KNOWLEDGE EXTRACTION IN RBES WHICH 
UNDERLYING LOGIC IS MPL 

 If the underlying logic of a RBES is Boolean logic, what 
can be asked is whether a formula can be obtained by for-
ward firing from the facts, rules and IC or not.  
 If the underlying logic is MPL, the certainty with which a 
formula can be obtained from the facts, rules and IC can be 
asked too (as done above for only one implication). 

Example 12 (Part I) 

 Let us consider now as propositional variables x, y, z, p, 
q, r, s, u, v, w. The only difference w.r.t. to the implementa-
tion used above is that, in the beginning, its second line of 
code will be  
> SV:=x,y,z,p,q,r,s,u,v,w: 

 Let the rules be, for instance: 
R1: x ∧¬ y → z 

R2: z → p∨u 
R3: v → ¬u 
R4: p → w 
R5: r → ¬w 
that is, in Maple: 
> R1:=(x &AND NEG(y))&IMP z: 

> R2:= z             &IMP(p &OR u): 

> R3:= v             &IMP NEG(u): 

> R4:= p             &IMP w: 

> R5:= r             &IMP NEG(w): 

 Then, the potential facts are: x, y, v, r (and their nega-
tions). 
 The ideal of rules and IC (there are no IC in this case) is: 
> iK:=[NEG(R1),NEG(R2),NEG(R3), 

       NEG(R4),NEG(R5)]; 

iK := [x-x*z-x*y+x*y*z, z-z*u- 
z*p+z*p*u, u*v, p-p*w, r*w] 

 A consistent set of facts is {x, ¬y, v}, and the correspon-
ding ideal is:  
> iJ:=[NEG(x),NEG(NEG(y)),NEG(v)]; 

        iJ = [1-x, y, 1-v] 

 The GB of I+J+K with respect to the variable order in 
sequence SV and pure lexicographical ordering for monomi-
als is: 
> base:=Basis([op(iI),op(iJ), 

             op(iK)], plex(SV));  
[-1+w, v^2-v, u*v, u^2-u, 
s^2-s, r^2-r, q^2-q, p-1, 

          -1+z, -1+y, -1+x] 

 Let us exit the example for a moment. 
 We can now perform knowledge extraction two ways.  
 One is using GB in a “booleanized” environment: truth 
values strictly greater than 0 are upgraded to 1 and 0s are left 
as they are. This is a only sufficient condition10 for a formula 
to follow by forward firing (without determining truth val-
ues).  

Example 12 (Part II) 

 In this particular RBES, if x, ¬y, v are the given facts, 
and facts, rules and IC with a truth value strictly greater than 
0 are associated the truth value 1, we have: from R1 and R2, 
p ∨ u is obtained, but from R3 we have ¬u, so we have p, 
and then, by R4, w is obtained.  
 This can be checked in the same RBES using classical 
Boolean logic: 
> NormalForm(NEG(p),base,plex(SV)); 

                 0 

> NormalForm(NEG(w),base,plex(SV)); 

                 0 

 The interpretation of these outputs (in continuous logic) 
would be that the formulae p and w can be reached by for-
ward firing (with a non null truth value, not determined).  
 The “booleanization” process could be improved in case 
each propositional variable always appeared in the antece-
dents of the rules either preceded by ¬ or not, but not in 
some rules one way and in other rules the other way (unfor-
tunately this is not always the case). Those preceded by ¬ 
and with truth values in (0,1) could be assigned the truth 
value 0 and those not preceded by ¬ and with truth values in 
(0,1) could be assigned the truth value 1.  
 The second method for knowledge extraction considered 
is straightforwardly obtained from the algebraic-numeric 
interpretation of formulae in continuous logic above and can 
precise the numerical truth values of the different formulae. 
Unfortunately, it has a drawback: substituting numerical val-
ues in the polynomial equations can lead to a non compatible 
system. This is not a logic inconsistency but a numerical one. 
Let us illustrate with a trivial example. 
Example 13 
 Let us consider the trivial RBES 
R1: x → z 
R2: y → z 
                                                
10It is not a necessary condition, as shown in the following example. If the truth values 
of x and y are 0.9 and 0.1, respectively, then x ∧¬y has truth value 0.9·(1-0.1)=0.81, 
meanwhile, if booleanized as suggested, 1·(1-1)=0 would be obtained. 

 In the Boolean case, we have that either none or some of 
the two rules can be fired. If none can be fired, z is not ob-
tained by forward firing. If exactly one or both can be fired, z 
is reached by forward firing. 
 Let us consider now the same rules in the continuous 
case. In the particular example, for instance, that the truth 
values of x, y, R1 and R2 are 8/10, 2/10, 7/10, 85/100 (re-
spectively), solving the two linear equations for z, we would 
have: 
> solve(subs(x=8/10,R1)=7/10,z); 

               5/8 

> solve(subs(y=2/10,R2)=85/100,z); 

               1/4 

that is, z=5/8 and z=1/4, that, from the algebraic point of 
view, are incompatible. 
 One possibility would be to allow only RBES where this 
did not happen. But this often happens. For instance in a 
RBES about car care, different rules lead to “stop engine 
immediately”. 
 Another option, if we wanted to precise the truth values 
of all formulae, would be to proceed step by step performing 
numerically the forward firing and actualizing (at each step) 
the truth values obtained for each variable to its maximum. 
Unfortunately this is not a polynomial process.  

LOGIC INCONSISTENCY OF RBES WHICH 
UNDERLYING LOGIC IS MPL USING GB 

 We shall consider a “worse case consistency”: we shall 
say that the RBES is inconsistent if and only if a consistent 
set of facts can be found, from which, if stated as true, by 
forward firing the rules and considering the IC (all stated as 
true), any formula can be obtained. 

Example 14 

 Let us consider the trivial RBES formed by the two rules: 
R1: x ∧ y → z 
R2: w →¬z 
that is, in Maple: 
> R1:=(x &AND y) &IMP z:  

> R2:=        w  &IMP NEG(z):  

 There is inconsistency, because x, y, w are potential facts, 
and, if they are all true (i.e., if they have the truth value 1), as 
R1 and R2 are also true, we would have: 
> subs(x=1,y=1,t=1,R1)=1; 

               z=1 

> subs(x=1,y=1,t=1,R2)=1; 

                1-z=1 

that is an incompatible system, as 1=z=0 is obtained.  
 We can alternatively check it using ideals, computing the 
GB of the ideal I+K+J (i.e., the ideal generated by the poly-
nomials in I and the negation of the facts, rules and IC): 
> iK:=[NEG(x),NEG(y),NEG(t)];  

      iK = [1 - x, 1 - y, 1-w] 

> iJ:=[NEG(R1),NEG(R2)]; 

      iJ = [x y - x y z, z w] 

> Basis([op(iI),op(iK),op(iJ)],  

        plex(SV));  
               [1] 

Example 15 

 Let us consider again the simple RBES of Example 12, 
which rules were: 
R1: x ∧¬ y → z 
R2: z → p∨u 
R3: v → ¬u 
R4: p → w 
R5: r → ¬w 
 As above: 
> iK:=[NEG(R1),NEG(R2),NEG(R3), 

       NEG(R4),NEG(R5)]: 

and for the consistent set of facts {x, ¬y, v}:  
> iJ:=[NEG(x),NEG(NEG(y)),NEG(v)]: 

 This consistent set of facts does not lead to inconsistency 
(the following GB is not [1]): 
> Basis([op(iI),op(iJ),op(iK)],  

        plex(SV));  
     [w-1, -1+v, u, s^2-s, r, 
     q^2-q, -1+p, -1+z, y, -1+x] 

 Meanwhile, the consistent set of facts {x, ¬y, v, r} does 
lead to an inconsistency: 
> iJ:=[NEG(x),NEG(NEG(y)),NEG(v), 

       NEG(r)]: 
> Basis([op(iI),op(iJ),op(iK)],  

        plex(SV)); 
                [1] 

CONCLUSIONS 

 A surprisingly tiny implementation for performing effec-
tive computations in Minimal Polynomial Logic (MPL) that 
is based on algebraic methods (Gröbner bases) and can be 
implemented in any Computer Algebra System has been 
presented. We consider it very convenient. 
 Although this approach can be applied to perform kno-
wledge extraction and verification of general RBES whose 
underlying logic is MPL, it only provides partial results and 
sufficient conditions, unlike the Boolean and many-valued 
cases.  

ABBREVIATIONS  

ES = Expert System(s) 
KB = Knowledge Base 
IE = Inference Engine 
GUI = Graphic Users Interface 
RBES = Rule Based Expert System(s) 
IC = Integrity Constraint(s) 
CAS = Computer Algebra System(s) 
NF = Normal Form 
GB = Gröbner Basis/Bases 
MPL = Minimal Polynomial Logic 
PL0 =  Lowest Degree Polynomial Logic 

